



University of British Columbia
Electrical and Computer Engineering
Digital Systems and Microcomputers
CPEN312

Lecture 14: Microcomputer Integer Arithmetic I

Dr. Jesús Calviño-Fraga P.Eng.
Department of Electrical and Computer Engineering, UBC
Office: KAIS 3024
E-mail: jesusc@ece.ubc.ca
Phone: (604)-827-5387

March 14/16, 2017

Copyright © 2009-2017, Jesús Calviño-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Objectives

- Perform single-byte-unsigned arithmetic add and subtract operations.
- Perform multi-byte unsigned arithmetic add and subtract operations.
- Compare single-byte and multi-byte numbers (=, <, >)
- Perform single-byte and multi-byte signed arithmetic add and subtract operations.
- Perform Binary to BCD and BCD to Binary conversion.

Lecture 14: Microcomputer Integer Arithmetic I

2

Copyright © 2009-2017, Jesús Calviño-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Unsigned-byte (8 bit) addition

Adding decimal numbers, by hand:

$$\begin{array}{r} 44 \\ + 55 \\ \hline 99 \end{array}$$

LSD

Adding binary numbers, by hand:

$$\begin{array}{r} 00101100 \quad (44) \\ + 00110111 \quad (55) \\ \hline 01100011 \quad (99) \end{array}$$

LSB

`MOV A, #44`
`ADD A, #55`

Lecture 14: Microcomputer Integer Arithmetic I

3

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Unsigned-byte (8 bit) addition

Add the content of R0 and R1 and store it in R7:

`MOV A, R0`

`ADD A, R1`

`MOV R7, A`

Add the content of memory locations 50 and 51, and save the result in location 52:

`MOV A, 50`

`ADD A, 51`

`MOV 52, A`

Lecture 14: Microcomputer Integer Arithmetic I

4

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Unsigned-byte (8 bit) subtraction

Subtract decimal numbers, by hand:

$$\begin{array}{r} 55 \\ - 44 \\ \hline 11 \end{array}$$

LSD

Subtract binary numbers, by hand:

$$\begin{array}{r} 00101100 \quad (55) \\ - 00110111 \quad (44) \\ \hline 00001011 \quad (11) \end{array}$$

LSB

CLR C
MOV A, #55
SUBB A, #44

Lecture 14: Microcomputer Integer Arithmetic I

5

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Unsigned-byte (8 bit) subtraction

Subtract the content of R0 and R1 and store it in R7:

```
MOV A, R0
CLR C
SUBB A, R1
MOV R7, A
```

Subtract the content of memory locations 60 and 61, and save the result in register DPL:

```
MOV A, 60
CLR C
SUBB A, 61
MOV DPL, A
```

Lecture 14: Microcomputer Integer Arithmetic I

6

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Unsigned multi-byte addition / subtraction and the Carry / Borrow flag

- We can handle integers of more than 8 bits by cascading addition/subtraction operation and use the carry/borrow flag.
- The MCS-51 have instructions to add both with carry and without carry, but it can only subtract with borrow!

– ADD
– ADDC
– SUBB

Lecture 14: Microcomputer Integer Arithmetic I

7

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

16-bit addition

LSD

$$\begin{array}{r}
 2047 \\
 + 322 \\
 \hline
 2369
 \end{array}$$

High byte Carry Low byte

$$\begin{array}{r}
 00000111 \quad 11111111 \\
 + 00000000 \quad 01000010 \\
 \hline
 00001001 \quad 01000001
 \end{array}$$

Lecture 14: Microcomputer Integer Arithmetic I

8

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

16-bit addition

In MCS-51 Assembly language:

```
mov a, #low(2047)
add a, #low(322)
mov b, a ; save low result in reg. b
mov a, #high(2047)
addc a, #high(322)
```

'high' and 'low' are assembler directives that take the upper or lower 8 bits of a 16-bit number.

Lecture 14: Microcomputer Integer Arithmetic I

9

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

16-bit addition

In MCS-51 Assembly language:

```
mov a, #0FFH
add a, #42H
mov b, a ; save low result in reg. b
mov a, #03H
addc a, #01H
```

Same as the previous slide, but you need to convert the numbers 2047 and 322 to hex (or binary!) and write them down in the program. Windows 'calc' is very handy for this!

Lecture 14: Microcomputer Integer Arithmetic I

10

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

32-bit addition

- Add the two 32 bit numbers located at RAM memory addresses 30H and 34H and store the result at XRAM address 100H.

NOTE: For regular internal memory we use the **MOV** instruction. For expanded memory we use the **MOVX** instruction.

First try: brute force!

```
mov a, 30H
add a, 34H
mov dptr, #100H
movx @dptr, a
mov a, 31H
addc a, 35H
inc dptr
movx @dptr, a
```

```
mov a, 32H
addc a, 36H
inc dptr
movx @dptr, a
mov a, 33H
addc a, 37H
inc dptr
movx @dptr, a
```

32-bit addition

- Add the two 32 bit numbers located at RAM memory addresses 30H and 34H and store the result at XRAM address 100H.
- Now, let us use assembler directives and indirect addressing (@R0, @R1, and @DPTR) in a compilable program:

Lecture 14: Microcomputer Integer Arithmetic I

13

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

32-bit addition

```
; Add32.asm: shows how to add two 32-bit numbers at RAM  
; addresses 30H and 34H and place result at XRAM address  
; 100H
```

```
$MODDE2
```

```
org 0000H  
    ljmp myprogram
```

```
DSEG at 30H
```

```
Num1: DS    4  
Num2: DS    4
```

```
XSEG at 100H  
Result: DS    4
```

Lecture 14: Microcomputer Integer Arithmetic I

14

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

32-bit addition

```
CSEG
myprogram:
    mov SP, #07FH
; Make Num1=55555555H for testing
    mov R0, #Num1
L1:    mov @R0, #55H
        inc R0
        cjne R0, #Num1+4, L1
; Make Num1=66666666H for testing
    mov R1, #Num2
L2:    mov @R1, #66H
        inc R1
        cjne R1, #Num2+4, L2
; Initialize pointers
    mov R0, #Num1
    mov R1, #Num2
    mov DPTR, #Result
```

Lecture 14: Microcomputer Integer Arithmetic I

15

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

32-bit addition

```
    clr c
    mov R2, #4
; Add the bytes, one by one
L3:    mov A, @R0
        addc A, @R1
        movx @dptr, A
        inc R0
        inc R1
        inc dptr
        djnz R2, L3

; Done! Loop forever
forever:
    jmp forever
END
```

Lecture 14: Microcomputer Integer Arithmetic I

16

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

16-bit subtraction

$$\begin{array}{r}
 14096 \\
 - 4128 \\
 \hline
 9968
 \end{array}$$

LSD

High byte		Borrow			Low byte
0 0 1 1 0 1 1 1					0 0 0 1 0 0 0 0
- 0 0 0 1 0 0 0 0					- 0 0 1 0 0 0 0 0
0 0 1 0 0 1 1 0		1			1 1 1 1 0 0 0 0

Lecture 14: Microcomputer Integer Arithmetic I

17

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

16-bit subtraction

In MCS-51 Assembly language:

```

clr c ; the 'carry' is also the 'borrow'!
mov a, #low(14096)
subb a, #low(4128)
mov b, a ; save low result in register b
mov a, #high(3710H)
subb a, #high(1020H)
    
```

14096=3710H=0011011100010000B

4128=1020H=0001000000100000B

Lecture 14: Microcomputer Integer Arithmetic I

18

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

32-subtraction...

```
clr c
mov a, 30H
subb a, 34H
mov dptr, #100H
movx @dptr, a
mov a, 31H
subb a, 35H
inc dptr
movx @dptr, a
```

```
mov a, 32H
subb a, 36H
inc dptr
movx @dptr, a
mov a, 33H
subb a, 37H
inc dptr
movx @dptr, a
```

Lecture 14: Microcomputer Integer Arithmetic I

19

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

32-bit subtraction...

Simply modify the example given above for 32-bit addition:

```
clr c
mov R2, #4
; Subtract the bytes, one by one
L3:  mov A, @R0
     subb A, @R1
     movx @dptr, A
     inc R0
     inc R1
     inc dptr
     djnz R2, L3

; Done! Loop forever
forever:
     jmp forever
END
```

Lecture 14: Microcomputer Integer Arithmetic I

20

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Comparing numbers

- We use single or multi-byte subtraction to compare two numbers:
 - If the result of the subtraction is zero: minuend = subtrahend.
 - If the result of the subtraction is not zero: minuend \neq subtrahend.
 - If 'borrow' set: minuend < subtrahend.
 - If no 'borrow' set: minuend > subtrahend.
- To check, we use the 'zero' and 'carry' flags and the instructions: **JZ**, **JNZ**, **JC**, and **JNC**.

Lecture 14: Microcomputer Integer Arithmetic I

21

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Are two 16-bit numbers Equal?

A 16-bit number is stored in R0 and R1, where R0 is the LSB.
Write the assembly code to check if the stored number is 1000.

```
clr c
mov a, #low(1000)
subb a, R0
jnz NotEqual
mov a, #high(1000)
subb a, R1
jnz NotEqual
sjmp Equal
```

'NotEqual' and 'Equal' are labels somewhere else in the program.

Lecture 14: Microcomputer Integer Arithmetic I

22

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Number1 > Number2?

Number1 is stored in R0 and R1, where R0 is the LSB. Number2 is stored at addresses 48H and 49H, where 48H is the LSB.

```
clr c
mov a, 48H
subb a, R0
mov a, 49H
subb a, R1
jc MyLabel ; jump if number1 > number2
```

Lecture 14: Microcomputer Integer Arithmetic I

23

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Number1 < Number2?

Number1 is stored in R0 and R1, where R0 is the LSB. Number2 is stored at addresses 48H and 49H, where 48H is the LSB.

```
clr c
mov a, 48H
subb a, R0
mov a, 49H
subb a, R1
jnc MyLabel ; jump if number1 < number2
```

WRONG!

Is the carry set if number1 = number2?

On the other hand, this is a perfectly good
(Number1 <= Number2)

Lecture 14: Microcomputer Integer Arithmetic I

24

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Number1 < Number2?

Number1 is stored in R0 and R1, where R0 is the LSB. Number2 is stored at addresses 48H and 49H, where 48H is the LSB.

```
clr c
mov a, R0
subb a, 48H
mov a, R1
subb a, 49H
jc MyLabel ; jump if number1 < number2
```


Lecture 14: Microcomputer Integer Arithmetic I

25

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Signed integer representation

True magnitude number representation:



Easy to understand, HARD to work with!


Lecture 14: Microcomputer Integer Arithmetic I

26

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Signed integer representation

2's-complement number representation:



Very convenient to perform arithmetic!

Lecture 14: Microcomputer Integer Arithmetic I

27

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

2's Complement

To obtain the 2's complement of a number:

0 0 1 1 0 1 0 0 Original number: +52

1 1 0 0 1 0 1 1 1's complement

1 Add 1

1 1 0 0 1 1 0 0 2's complement

In 8051's assembly language:

MOV A, #52

CPL A

INC A

Or

CLR C

CLR A

SUBB A, #52

Lecture 14: Microcomputer Integer Arithmetic I

28

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Signed Integer Addition

Case 1: two positive numbers:

$$\begin{array}{r}
 0\ 0\ 1\ 1\ 0\ 1\ 1\ 1\ 55 \\
 +\ 0\ 0\ 0\ 1\ 0\ 1\ 0\ 0\ 20 \\
 \hline
 0\ 1\ 0\ 0\ 1\ 0\ 1\ 1\ 75
 \end{array}$$

Case 2: positive number larger than negative number:

$$\begin{array}{r}
 0\ 0\ 1\ 1\ 0\ 1\ 1\ 1\ 55 \\
 +\ 1\ 1\ 0\ 0\ 1\ 1\ 0\ 0\ -52 \\
 \hline
 0\ 0\ 0\ 0\ 0\ 0\ 1\ 1\ 3
 \end{array}$$

Lecture 14: Microcomputer Integer Arithmetic I

29

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Signed Integer Addition

Case 3: negative number larger than positive number :

$$\begin{array}{r}
 0\ 0\ 0\ 1\ 0\ 1\ 0\ 0\ 20 \\
 +\ 1\ 1\ 0\ 0\ 1\ 1\ 0\ 0\ -52 \\
 \hline
 1\ 1\ 1\ 0\ 0\ 0\ 0\ 0\ -32
 \end{array}$$

Case 4: Two negative numbers:

$$\begin{array}{r}
 1\ 1\ 1\ 1\ 0\ 0\ 0\ 1\ -15 \\
 +\ 1\ 1\ 0\ 0\ 1\ 1\ 0\ 0\ -52 \\
 \hline
 1\ 0\ 1\ 1\ 1\ 1\ 0\ 1\ -67
 \end{array}$$

Lecture 14: Microcomputer Integer Arithmetic I

30

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Signed Integer Addition

Case 5: equal and opposite numbers :

$$\begin{array}{r}
 00110100 \quad 52 \\
 + 11001100 \quad -52 \\
 \hline
 00000000 \quad 0
 \end{array}$$

This technique also works for decimal number calculations. It is particularly handy when performing subtraction by hand. For more information, check the "method of complements" in Wikipedia:

http://en.wikipedia.org/wiki/Method_of_complements

Lecture 14: Microcomputer Integer Arithmetic I

31

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Method of Complements for Hand Calculation of Decimal Subtraction

$$\begin{array}{r}
 985318031 \\
 - 45170246 \\
 \hline
 \end{array}$$

'Negate' in base 10 and add 1 to get the ten's complement

$$\begin{array}{r}
 954829753 \\
 + 1 \\
 \hline
 954829754
 \end{array}$$

Add the Ten's complement

$$\begin{array}{r}
 985318031 \\
 + 954829754 \\
 \hline
 1940147785
 \end{array}$$

Disregard!

Correct answer!

Lecture 14: Microcomputer Integer Arithmetic I

32

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Signed Integer Addition

Write the assembly code to add two 16-bit signed numbers at RAM addresses 20H-21H and 30H-31H and save the result in R0-R1.

```
mov A, 21H
add A, 31H
mov R1, A
mov A, 20H
addc A, 30H
mov R0, A
```

Lecture 14: Microcomputer Integer Arithmetic I

33

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Signed Integer Subtraction

Exactly the same as signed addition. Just get the 2's complement of the subtrahend before the addition!

Write the assembly code to subtract the signed 16-bit number at addresses 30H-31H from the 16-bit signed number at RAM addresses 20H-21H. Save the result in R0-R1.

<pre>mov A, 31H cpl A add A, #1 mov R1, A mov A, 30H cpl A addc A, #0 mov R0, A</pre>	<pre>mov A, 21H add A, R1 mov R1, A mov A, 20H addc A, R0 mov R0, A</pre>
---	---

Lecture 14: Microcomputer Integer Arithmetic I

34

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Binary to BCD Conversion

- Registers, counters, addresses, numbers, etc. in the microcontroller are represented in binary.
- People like to work with decimal numbers! Also, it is very easy to convert BCD to ASCII:

```
mov R1, #91H ; BCD number
mov a, R1
swap a
anl a, #0FH
orl a, #30H ; ASCII "9" is #39H
mov 50H, a
mov a, R1
anl a, #0FH
orl a, #30H ; ASCII "1" is #31H
mov 51H, a
```

Code used to represent English characters and symbols in displays, terminals, and printers

Lecture 14: Microcomputer Integer Arithmetic I

35

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

ASCII Table

Dec	Hx	Oct	Char	Dec	Hx	Oct	Html	Chr	Dec	Hx	Oct	Html	Chr	Dec	Hx	Oct	Html	Chr
0	0	000	NUL (null)	32	20	040	#32;	Space	64	40	100	#64;	@	96	60	140	#96;	`
1	1	001	SOH (start of heading)	33	21	041	#33;	!	65	41	101	#65;	A	97	61	141	#97;	a
2	2	002	STX (start of text)	34	22	042	#34;	"	66	42	102	#66;	B	98	62	142	#98;	b
3	3	003	ETX (end of text)	35	23	043	#35;	#	67	43	103	#67;	C	99	63	143	#99;	c
4	4	004	EOT (end of transmission)	36	24	044	#36;	\$	68	44	104	#68;	D	100	64	144	#100;	d
5	5	005	ENQ (enquiry)	37	25	045	#37;	%	69	45	105	#69;	E	101	65	145	#101;	e
6	6	006	ACK (acknowledge)	38	26	046	#38;	&	70	46	106	#70;	F	102	66	146	#102;	f
7	7	007	BEL (bell)	39	27	047	#39;	'	71	47	107	#71;	G	103	67	147	#103;	g
8	8	010	BS (backspace)	40	28	050	#40;	(72	48	110	#72;	H	104	68	150	#104;	h
9	9	011	TAB (horizontal tab)	41	29	051	#41;)	73	49	111	#73;	I	105	69	151	#105;	i
10	A	012	LF (NL line feed, new line)	42	2A	052	#42;	*	74	4A	112	#74;	J	106	6A	152	#106;	j
11	B	013	VT (vertical tab)	43	2B	053	#43;	+	75	4B	113	#75;	K	107	6B	153	#107;	k
12	C	014	FF (NP form feed, new page)	44	2C	054	#44;	,	76	4C	114	#76;	L	108	6C	154	#108;	l
13	D	015	CR (carriage return)	45	2D	055	#45;	-	77	4D	115	#77;	M	109	6D	155	#109;	m
14	E	016	SO (shift out)	46	2E	056	#46;	.	78	4E	116	#78;	N	110	6E	156	#110;	n
15	F	017	SI (shift in)	47	2F	057	#47;	/	79	4F	117	#79;	O	111	6F	157	#111;	o
16	10	020	DLE (data link escape)	48	30	060	#48;	0	80	50	120	#80;	P	112	70	160	#112;	p
17	11	021	DC1 (device control 1)	49	31	061	#49;	1	81	51	121	#81;	Q	113	71	161	#113;	q
18	12	022	DC2 (device control 2)	50	32	062	#50;	2	82	52	122	#82;	R	114	72	162	#114;	r
19	13	023	DC3 (device control 3)	51	33	063	#51;	3	83	53	123	#83;	S	115	73	163	#115;	s
20	14	024	DC4 (device control 4)	52	34	064	#52;	4	84	54	124	#84;	T	116	74	164	#116;	t
21	15	025	NAK (negative acknowledge)	53	35	065	#53;	5	85	55	125	#85;	U	117	75	165	#117;	u
22	16	026	SYN (synchronous idle)	54	36	066	#54;	6	86	56	126	#86;	V	118	76	166	#118;	v
23	17	027	ETB (end of trans. block)	55	37	067	#55;	7	87	57	127	#87;	W	119	77	167	#119;	w
24	18	030	CAN (cancel)	56	38	070	#56;	8	88	58	130	#88;	X	120	78	170	#120;	x
25	19	031	EM (end of medium)	57	39	071	#57;	9	89	59	131	#89;	Y	121	79	171	#121;	y
26	1A	032	SUB (substitute)	58	3A	072	#58;	:	90	5A	132	#90;	Z	122	7A	172	#122;	z
27	1B	033	ESC (escape)	59	3B	073	#59;	;	91	5B	133	#91;	[123	7B	173	#123;	{
28	1C	034	FS (file separator)	60	3C	074	#60;	<	92	5C	134	#92;	\	124	7C	174	#124;	
29	1D	035	GS (group separator)	61	3D	075	#61;	=	93	5D	135	#93;]	125	7D	175	#125;	}
30	1E	036	RS (record separator)	62	3E	076	#62;	>	94	5E	136	#94;	^	126	7E	176	#126;	~
31	1F	037	US (unit separator)	63	3F	077	#63;	?	95	5F	137	#95;	_	127	7F	177	#127;	DEL

Binary to BCD conversion of 8-bit numbers in the 8051

```
; Eight bit number to convert passed in acc.
; Result in r1,r0
Bin2BCD_8bit:
    mov b, #100
    div ab
    mov r1, a    ; Save hundreds
    mov a, b     ; Remainder is in register b
    mov b, #10
    div ab
    swap a
    orl a, b
    mov r0, a    ; Save tens and ones
    ret
```

Lecture 14: Microcomputer Integer Arithmetic I

37

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Binary to BCD Conversion: the double dabble algorithm

1. BIN: 01011011 BCD=000
Multiply BDC by two and add the underlined bit:
 $BCD=(000+000+0)=000$
2. BIN: 01011011 BCD=000
Multiply BDC by two and add the underlined bit:
 $BCD=(000+000+1)=001$
3. BIN: 01011011 BCD=001
Multiply BDC by two and add the underlined bit:
 $BCD=(001+001+0)=002$
4. BIN: 01011011 BCD=002
Multiply BDC by two and add the underlined bit:
 $BCD=(002+002+1)=005$

Lecture 14: Microcomputer Integer Arithmetic I

38

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Binary to BCD Conversion : the double dabble algorithm

5. BIN: 01011011 BCD=005
Multiply BDC by two and add the underlined bit:
 $BCD=(005+005+1)=011$
6. BIN: 01011011 BCD=011
Multiply BDC by two and add the underlined bit:
 $BCD=(011+011+0)=022$
7. BIN: 01011011 BCD=022
Multiply BDC by two and add the underlined bit:
 $BCD=(022+022+1)=045$
8. BIN: 01011011 BCD=045
Multiply BDC by two and add the underlined bit:
 $BCD=(045+045+1)=091$ ← Final result is 91

Lecture 14: Microcomputer Integer Arithmetic I

39

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Multiply a BCD number by 2.

- If the processor supports addition of BCD numbers then add the BCD number to itself.
- If the processor DOES NOT support addition of BCD numbers:
 - Add 3 (0011B) to each BCD digit > 4 (0100B).
 - Shift Left one bit.

Lecture 14: Microcomputer Integer Arithmetic I

40

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

BCD addition in the 8051: the “DA A” instruction.

- **Function:** Decimal Adjust Accumulator
- **Description:** DA adjusts the contents of the Accumulator to correspond to a BCD (Binary Coded Decimal) number after two BCD numbers have been added by the ADD or ADDC instruction. If the carry bit is set or if the value of bits 0-3 exceed 9, 06H is added to the accumulator. If the carry bit was set when the instruction began, or if 06H was added to the accumulator in the first step, 60H is added to the accumulator.

Page 61 of course textbook has more details on the works of this instruction!

Lecture 14: Microcomputer Integer Arithmetic I

41

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

How to multiply a BCD by two if BCD addition is not available.

Multiply BCD number 3925 by 2:

```
0011 1001 0010 0101  BCD number to multiply by 2
0000 0011 0000 0011  Correction before left shift
0011 1100 0010 1000  Sum
0111 1000 0101 0000  Shift left
  7   8   5   0
```

$$3925 * 2 = 7850$$

Lecture 14: Microcomputer Integer Arithmetic I

42

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

16-bit binary to BCD

```
;Converts the hex number in R0-R1 to BCD
;in R2-R3-R4
hex2bcd:
    mov R2, #0 ;Set BCD result to 00000
    mov R3, #0
    mov R4, #0
    mov R5, #16 ;Loop counter.
L0:
    mov a, R1 ;Shift R0-R1 left through carry
    rlc a
    mov R1, a
    mov a, R0
    rlc a
    mov R0, a
```

Continues...

Lecture 14: Microcomputer Integer Arithmetic I

43

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

16-bit binary to BCD

```
; Perform bcd + bcd + carry
; using BCD numbers
mov a, R4
addc a, R4
da a
mov R4, a
mov a, R3
addc a, R3
da a
mov R3, a
mov a, R2
addc a, R2
mov R2, a
djnz R5, L0
ret
```

Lecture 14: Microcomputer Integer Arithmetic I

44

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

BCD to Binary Conversion

- We may need to convert to binary any number provided by humans!
- Many chips work only with BCD numbers, for example Real Time Clocks (RTCs) often count in BCD only.
- One way to convert BCD to binary is by multiplying by ten (10) and adding a four bit number:

Lecture 14: Microcomputer Integer Arithmetic I

45

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

BCD to Binary Conversion for 8-bit binary numbers in the 8051.

1. BCD=137 BIN: 00000000
Multiply BIN by ten and add the underlined digit:
$$\text{BIN} = ((00000000 * 1010) + 0001) = 00000001$$
2. BCD=137 BIN: 00000001
Multiply BIN by ten and add the underlined digit:
$$\text{BIN} = ((00000001 * 1010) + 0011) = 00001101$$
3. BCD=137 BIN: 00001101
Multiply BIN by ten and add the underlined digit:
$$\text{BIN} = ((00001101 * 1010) + 0111) = \mathbf{10001001}$$

Lecture 14: Microcomputer Integer Arithmetic I

46

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

BCD to Binary Conversion for 8-bit binary numbers in the 8051.

- Convert a three digit BCD number (0 to 255) stored in R6-R7 to binary and save the result in R5.

```

mov a, R6
mov b, #10
mul ab
mov R5, a
mov a, R7
swap a
anl a, #0fH
add a, R5
mov b, #10
mul ab
mov R5, a
    
```

```

mov a, R7
anl a, #0fH
add a, R5
mov R5, a
    
```

Lecture 14: Microcomputer Integer Arithmetic I

47

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

BCD to Binary Conversion for n-bit binary numbers.

Convert 205 to binary:

```

205 ÷ 2 = 102 R = 1
102 ÷ 2 = 51 R = 0
51 ÷ 2 = 25 R = 1
25 ÷ 2 = 12 R = 1
12 ÷ 2 = 6 R = 0
6 ÷ 2 = 3 R = 0
3 ÷ 2 = 1 R = 1
1 ÷ 2 = 0 R = 1
    
```

To convert BCD to binary we need to divide a BCD number by two! This algorithm works in reverse from the double dabble algorithm!!!

205 = 11001101B = 0CDH

Lecture 14: Microcomputer Integer Arithmetic I

48

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Divide BCD by two

- Shift the BCD number right by one
- If the most significant bit of a BCD digit is one, subtract 3 to that digit.
- The least significant bit of the least significant BCD digit is the remainder.

Lecture 14: Microcomputer Integer Arithmetic I

49

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Divide BCD by two

Divide BCD number 3925 by 2:

0011 1001 0010 0101 BCD number to divide by 2

0001 1100 1001 0010 Right shift

0000 0011 0011 0000 Correction

0001 1001 0110 0010 Subtract

1 9 6 2

3925 / 2 = 1962, remainder=1

Lecture 14: Microcomputer Integer Arithmetic I

50

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

16-bit BCD to binary

```
; bcd2hex: Converts the BCD number in R2-R3-R4
; to binary in R0-R1

rrc_and_correct:
    rrc a
    mov r6, psw ; Save carry (it is changed by the add)
    jnb acc.7, nocor1
    add a, #(100H-30H) ; subtract 3 from packed BCD MSD
nocor1:
    jnb acc.3, nocor2
    add a, #(100H-03H) ; subtract 3 from packed BCD LSD
nocor2:
    mov psw, r6 ; Restore carry
    ret
```

Lecture 14: Microcomputer Integer Arithmetic I

51

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

16-bit BCD to binary

```
bcd2hex:
    mov R5, #16 ;Loop counter.
L0:
    ; Divide BCD by two
    clr c
    mov a, R2
    lcall rrc_and_correct
    mov R2, a
    mov a, R3
    lcall rrc_and_correct
    mov R3, a
    mov a, R4
    lcall rrc_and_correct
    mov R4, a
```

Lecture 14: Microcomputer Integer Arithmetic I

52

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

16-bit BCD to binary

```
;Shift R0-R1 right through carry
mov a, R0
rrc a
mov R0, a
mov a, R1
rrc a
mov R1, a

djnz R5, L0
ret
```

Lecture 14: Microcomputer Integer Arithmetic I

53

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.

Exercises

- Write an assembly subroutine for the 8051 that checks if a 32-bit number stored in registers R0 to R3 is zero.
- Write the assembly equivalent of this piece of C code (the size of int is 2 bytes):
 unsigned int x, y;
 unsigned char z;
 .
 [other code comes here]

```
.
if (x>y) z=0;
else z=1;
```

Assembly for this!

Lecture 14: Microcomputer Integer Arithmetic I

54

Copyright © 2009-2017, Jesus Calvino-Fraga. Not to be copied, used, or revised without explicit written permission from the copyright owner.